clear
set more off
set matsize 800
set mem 800m
version 10.0

*** Aufzeichung in Protokoll starten.
capture log close

*** Pfad fuer lokale Testlaeufe
global root "/Users/sascha/work/Ludger/texdocs/suicide/Todesursachenstatistik/"
*** Pfad beim FDZ Sachsen
*global root "D:/data/23211 Todesursachenstatistik/Nutzungen/1813-2011 ifo Muenchen - Knoche/"
log using "$root/Analysen/20121004/20121004_tus-analyse.log", replace

*** Titel des Projekts: Analyse des Einflusses von Religionszugehoerigkeit auf natuerliche und gewaltsame Todesursachen 
*** Datengrundlage:	<Todesursachenstatistik> 
*** 
*** Dateiname des Programmcodes: tus-analyse.do
*** erstellt: 06.08.2012
*** von: Sascha O. Becker und Ludger Woessmann
*** E-Mail: sascha.becker@gmx.de
*** Tel. 0044 7878 7578 11
*** 
*** Dateiname des Output-Files: tus-analyse.log 
*** 
*** 
*** Grundriss des Programms: Untersuchung von Religionszugehoerigkeit und Selbstmord
*** 
*** Verwendete Variablen:
*** Originalvariablen:  
*** ef304             Wohnort der/des Verstorbenen
*** ef305u3           Sterbedatum - Jahr
*** ef306             Geschlecht
*** ef310             Familienstand
*** ef313             Religionszugehoerigkeit
*** ef314             Staatsangehoerigkeit
*** ef316             Todesursache (ICD-Schluessel)
*** ef317             Todesursache (Ursache der Schaedigung)
*** ef320u3           Sterbealter - in Jahren
***
*** Neu angelegte Variablen: 	Stetige Variablen:
*** ags               Amtlicher Gemeindeschluessel Wohnort der/des Verstorbenen numerisch statt string
*** age               Sterbealter
*** age2              Sterbealter^2 / 100
*** Dummy-Variablen:
*** male              maennlich
*** famstand_unknown  Familienstand unbekannt 
*** famstand_single   Familienstand Single
*** famstand_married  Familienstand verheiratet
*** famstand_widowed  Familienstand verwitwet 
*** famstand_divorced Familienstand geschieden
*** rel_unknown       Religionszugehoerigkeit unbekannt
*** rel_protestant    Religionszugehoerigkeit protestantisch
*** rel_catholic      Religionszugehoerigkeit katholisch oder orthodox
*** rel_otherchrist   Religionszugehoerigkeit sonstige christliche
*** rel_jewish        Religionszugehoerigkeit juedisch
*** rel_islam         Religionszugehoerigkeit muslimisch
*** rel_other         Religionszugehoerigkeit sonstige
*** rel_none          Religionszugehoerigkeit gemeinschaftslos
*** rel_pro_no        Religionszugehoerigkeit protestantisch oder gemeinschaftslos
*** german            Staatsangehoerigkeit Deutsch
*** suicide           Selbstmord-Dummy
***
*** auf Kreisebene (ags) aggregierte Variablen, die als Kontrollvariablen herangezogen werden
*** n_protestant_ags: Anzahl Protestanten an den Toten im Landkreis ags
*** n_catholic_ags: Anzahl Katholiken an den Toten im Landkreis ags
*** prot_in_protcath_ags: Anteil Protestanten an den verstorbenen Katholiken und Protestanten des Landkreises ags
***
*** Variablen aus der VZ1987 auf Landkreisebene:
*** log_pop_tot_res_t             log(Bevoelkerungszahl im Landkreis)
*** prot_in_protcath              Protestantenanteil (an Christen)
*** share_pop_under3_res_t        Bevoelkerungsanteil Alter 0-2
*** share_pop_3_5_res_t           Bevoelkerungsanteil Alter 3-4
***    etc etc
*** share_pop_70_75_res_t         Bevoelkerungsanteil Alter 70-74
*** share_pop_75andmore_res_t     Bevoelkerungsanteil Alter 75 plus
*** share_pop_tot_res_single_t    Anteil Singles an Gesamtbevoelkerung
*** share_pop_tot_res_married_t   Anteil Verheiratete an Gesamtbevoelkerung
*** share_pop_tot_res_wid_t       Anteil Witwen/Witwer an Gesamtbevoelkerung
*** share_pop_tot_res_div_t       Anteil Geschiedene an Gesamtbevoelkerung
*** share_foreigners              Auslaenderanteil an Gesamtbevoelkerung
*** share_pop_res_wrkmninc_1st_t  Anteil Gesamtbevoelkerung, die arbeitet in Land- und Forstwirtschaft, Fischerei
*** share_pop_res_wrkmninc_2nd_t  Anteil Gesamtbevoelkerung, die arbeitet in Produzierendes Gewerbe
*** share_pop_res_wrkmninc_3rd_t  Anteil Gesamtbevoelkerung, die arbeitet in Handel, Verkehr und Nachrichtenuebermittlung
*** share_pop_res_wrkmninc_oth_t  Anteil Gesamtbevoelkerung, die arbeitet in uebrige Wirtschaftsbereiche
*** share_pop_res_incsub_t        Anteil Gesamtbevoelkerung, die lebt von Arbeitslosengeld/-hilfe, Rente, Pension
*** share_pop_res_incpar_t        Anteil Gesamtbevoelkerung, die lebt von Unterhalt durch Eltern, Ehegatten usw.
*** share_school_prim             Anteil Gesamtbevoelkerung mit Volksschule, Hauptschule
*** share_school_lowsec           Anteil Gesamtbevoelkerung mit Realschule oder gleichwertiger Abschluss 
*** share_school_uppsec           Anteil Gesamtbevoelkerung mit Hochschul-/ Fachhochschulreife
*** share_voc_fach                Anteil Gesamtbevoelkerung mit Berufsfach-/ Fachschule
*** share_voc_uni                 Anteil Gesamtbevoelkerung mit Hochschule / Fachhochschule
*** dist_wittenberg               Entfernung nach Wittenberg

**** Lege Arbeitsverzeichnis an
*capture mkdir "$root/Analysen/20120619"

*** Erlaube Durchlauf des Programms
set more off

/*******************/
/* 1992-1997: ICD9 */
/*******************/

*** Datensatz einlesen
use "$root/KDFV/tus080_1992.dta", clear

*** Generiere Amtlichen Gemeindeschluessel Wohnort der/des Verstorbenen numerisch statt string
gen ags=real(ef304)

*** Generiere Dummyvariable fuer Maenner
gen male=(ef306==1)

*** Generiere Dummyvariablen fuer Familienstand
gen famstand_unknown  = (ef310==0 | ef310==9) 
gen famstand_single   = (ef310==1)
gen famstand_married  = (ef310==2) /* reference category */
gen famstand_widowed  = (ef310==3)
gen famstand_divorced = (ef310==4)

*** Generiere Dummyvariablen fuer Religionszugehoerigkeit
gen rel_unknown = (ef313==0) 
gen rel_protestant = (ef313==1 | ef313==2)
gen rel_catholic = (ef313==3 | ef313==4 | ef313==5) /* reference category */
gen rel_otherchrist = (ef313==6)
gen rel_jewish = (ef313==7)
gen rel_islam = (ef313==8)
gen rel_other = (ef313==9 | ef313==10)
gen rel_none  = (ef313==11)
gen rel_pro_no = rel_protestant + rel_none

*** Protestantenanteil an "Protestanten plus Katholiken" auf Landkreisebene aus TUS
bysort ags: egen n_protestant_ags = sum(rel_protestant)
bysort ags: egen n_catholic_ags = sum(rel_catholic)
bysort ags: gen prot_in_protcath_ags = n_protestant_ags/(n_catholic_ags+n_protestant_ags)

*** Deutsche vs Auslaender
gen german = (ef314==0)

*** Generiere Selbstmord-Dummy
*** Selbstmord-Codes in ICD9 sind Codes 950 bis 959 in den FDZ-Daten, aber E950 bis E959 in ICD9-Doku, d.h. 10 Codes
gen suicide=0
foreach num of numlist 950/959 {
  replace suicide=1 if ef317=="`num'"
}

*** Generiere alter^2 auf Basis von Alter: Sterbealter - in Jahren
rename ef320u3 age
gen age2 = age * age / 100

*** Sortiere nach AGS
sort ags

*** Verknuepfe mit Kreisdaten der VZ 1987
merge ags using "$root/KDFV/VZ1987.dta"

*** Benenne Variable Protestantenanteil um
ren share_prot_in_protcath prot_in_protcath

*** Betrachte Korrelation zwischen Protestantenanteil in TUS und Protestantenanteil in VZ1987
pwcorr prot_in_protcath prot_in_protcath_ags, sig

*** Generiere log(Bevoelkerungszahl) im Landkreis
gen log_pop_tot_res_t = log(pop_tot_res_t)

*** Tabelliere Religionszugehoerigkeit der Verstorben insgesamt, sowie fuer die Selbstmoerder
*tab ef313
*tab ef313 if suicide==1



*** Regressionen: was "erklaert" Selbstmord als Todesart

*** (a) nur Variablen auf Individualebene
/************/
/* COLUMN 1 */ 
/************/
regress suicide rel_protestant rel_otherchrist rel_jewish rel_islam rel_other rel_none rel_unknown german age age2 male famstand_unknown famstand_single famstand_widowed famstand_divorced 

* Kontrolliere fuer fixe Landkreiseffekte (Variante 1 via -absorb-)
/************/
/* COLUMN 2 */ 
/************/
areg    suicide rel_protestant rel_otherchrist rel_jewish rel_islam rel_other rel_none rel_unknown german age age2 male famstand_unknown famstand_single famstand_widowed famstand_divorced, absorb(ags)

* nur f�r Westdeutschland (da nur dort VZ 1987 verf�gbar)
/************/
/* COLUMN 5 */ 
/************/
areg    suicide rel_protestant rel_otherchrist rel_jewish rel_islam rel_other rel_none rel_unknown german age age2 male famstand_unknown famstand_single famstand_widowed famstand_divorced log_pop_tot_res_t share*, absorb(ags) vce(cluster ags)
/************/
/* COLUMN 6 */ 
/************/
regress suicide rel_protestant rel_otherchrist rel_jewish rel_islam rel_other rel_none rel_unknown german age age2 male famstand_unknown famstand_single famstand_widowed famstand_divorced log_pop_tot_res_t share*, cluster(ags)
/************/
/* COLUMN 7 */ 
/************/
regress suicide prot_in_protcath rel_otherchrist rel_jewish rel_islam rel_other rel_none rel_unknown german age age2 male famstand_unknown famstand_single famstand_widowed famstand_divorced log_pop_tot_res_t share*, cluster(ags)
/************/
/* COLUMN 8 */ 
/************/
regress suicide rel_protestant prot_in_protcath rel_otherchrist rel_jewish rel_islam rel_other rel_none rel_unknown german age age2 male famstand_unknown famstand_single famstand_widowed famstand_divorced log_pop_tot_res_t share*, cluster(ags)

/*------------------------------------------------------------------------------------------------------*/
/*------------------------------------------------------------------------------------------------------*/
/*------------------------------------------------------------------------------------------------------*/

/********************/
/* 1998-2009: ICD10 */
/********************/

**** Datensatz einlesen
use "$root/KDFV/tus080_2009.dta", clear

*** Generiere Amtlichen Gemeindeschluessel Wohnort der/des Verstorbenen numerisch statt string
gen ags=real(ef304)

*** Generiere Dummyvariable fuer Maenner
gen male=(ef306==1)

*** Generiere Dummyvariablen fuer Familienstand
gen famstand_unknown  = (ef310==0 | ef310==9) 
gen famstand_single   = (ef310==1)
gen famstand_married  = (ef310==2) /* reference category */
gen famstand_widowed  = (ef310==3)
gen famstand_divorced = (ef310==4)

*** Generiere Dummyvariablen fuer Religionszugehoerigkeit
gen rel_unknown = (ef313==0) 
gen rel_protestant = (ef313==1 | ef313==2)
gen rel_catholic = (ef313==3 | ef313==4 | ef313==5) /* reference category */
gen rel_otherchrist = (ef313==6)
gen rel_jewish = (ef313==7)
gen rel_islam = (ef313==8)
gen rel_other = (ef313==9 | ef313==10)
gen rel_none  = (ef313==11)
gen rel_pro_no = rel_protestant + rel_none

*** Protestantenanteil an "Protestanten plus Katholiken" auf Landkreisebene aus TUS
bysort ags: egen n_protestant_ags = sum(rel_protestant)
bysort ags: egen n_catholic_ags = sum(rel_catholic)
bysort ags: gen prot_in_protcath_ags = n_protestant_ags/(n_catholic_ags+n_protestant_ags)

*** Deutsche vs Auslaender
gen german = (ef314==0)

**** Generiere Selbstmord-Dummy
*** Selbstmord-Codes in ICD10 sind Codes X60, X61, ..., X84, d.h. 25 Codes */
gen suicide=0
foreach num of numlist 60/84 {
  replace suicide=1 if ef317=="X`num'"
}

*** Generiere alter^2 auf Basis von Alter: Sterbealter - in Jahren
rename ef320u3 age
gen age2 = age * age / 100

*** Sortiere nach AGS
sort ags

*** Verknuepfe mit Kreisdaten der VZ 1987
merge ags using "$root/KDFV/VZ1987.dta"

*** Benenne Variable Protestantenanteil um
ren share_prot_in_protcath prot_in_protcath

*** Betrachte Korrelation zwischen Protestantenanteil in TUS und Protestantenanteil in VZ1987
pwcorr prot_in_protcath prot_in_protcath_ags, sig

*** Generiere log(Bevoelkerungszahl) im Landkreis
gen log_pop_tot_res_t = log(pop_tot_res_t)

*** Tabelliere Religionszugehoerigkeit der Verstorben insgesamt, sowie fuer die Selbstmoerder
*tab ef313
*tab ef313 if suicide==1



*** Regressionen: was "erklaert" Selbstmord als Todesart

*** (a) nur Variablen auf Individualebene
/************/
/* COLUMN 3 */ 
/************/
regress suicide rel_protestant rel_otherchrist rel_jewish rel_islam rel_other rel_none rel_unknown german age age2 male famstand_unknown famstand_single famstand_widowed famstand_divorced 

* Kontrolliere fuer fixe Landkreiseffekte (Variante 1 via -absorb-)
/************/
/* COLUMN 4 */ 
/************/
areg suicide rel_protestant rel_otherchrist rel_jewish rel_islam rel_other rel_none rel_unknown german age age2 male famstand_unknown famstand_single famstand_widowed famstand_divorced, absorb(ags)

log close
